Ecuaciones Diferenciales
grupo 15 semestre 2023-1
Segundo Examen Parcial: Temas 3 & 4
SOLUCIÓN

2023-11-23

PREGUNTA 1 (20 puntos) Mediante la Transformada de Laplace obtenga la solución de la ecuación diferencial, sujeta a las condiciones iniciales dadas (***sin usar dsolve***)

```
> restart
> Ecua := 3 (diff(y(t), t$2)) + y(t) = sin(t) Heaviside(t - 2 π)
```

$$Ecua := 3\left(\frac{d^2}{dt^2}\,y(t)\right) + y(t) = \sin(t)\,\mathrm{Heaviside}(t - 2\pi) \quad \textbf{(1)}$$

```
> Cond := y(0) = 1, D(y)(0) = 0
```

$$Cond := y(0) = 1, D(y)(0) = 0 \quad \textbf{(2)}$$

```
> with(inttrans) :
> EcuaTransLap := subs(Cond, laplace(Ecua, t, s))
```

$$EcuaTransLap := 3\,s^2\,laplace(y(t), t, s) - 3\,s + laplace(y(t), t, s) = \frac{e^{-2\,s\,\pi}}{s^2 + 1} \quad \textbf{(3)}$$

```
> SolTransLap := simplify(isolate(EcuaTransLap, laplace(y(t), t, s)))
```

$$SolTransLap := laplace(y(t), t, s) = \frac{3\,s^3 + e^{-2\,s\,\pi} + 3\,s}{(s^2 + 1)\,(3\,s^2 + 1)} \quad \textbf{(4)}$$

```
> SolPart := invlaplace(SolTransLap, s, t)
```

$$SolPart := y(t) = \cos\left(\frac{1}{3}\sqrt{3}\,t\right) + \frac{1}{2}\left(-\sin(t) + \sqrt{3}\,\sin\left(\frac{1}{3}\sqrt{3}\,(t - 2\pi)\right)\right)\mathrm{Heaviside}(t - 2\pi) \quad \textbf{(5)}$$

```
> Comprob := simplify(eval(subs(y(t) = rhs(SolPart), lhs(Ecua) - rhs(Ecua) = 0)))
```

$$Comprob := 0 = 0 \quad \textbf{(6)}$$

```
> plot(rhs(SolPart), t = 0 ..6·Pi)
```



> *restart*

PREGUNTA 2 (20 puntos) Obtener la solución particular del sistema de ecuaciones diferenciales con las condiciones iniciales dadas (***sin usar dsolve***)

> *SistEcua* := *diff*(y[1](t), t) = y[2](t), *diff*(y[2](t), t) = −4·y[1](t) + 2·cos(t) : *SistEcua*[1]; *SistEcua*[2];

$$\frac{d}{dt} y_1(t) = y_2(t)$$

$$\frac{d}{dt} y_2(t) = -4\, y_1(t) + 2\cos(t) \quad \textbf{(7)}$$

> *Cond* := y[1](0) = 0, y[2](0) = 0

$$Cond := y_1(0) = 0, y_2(0) = 0 \quad \textbf{(8)}$$

> *AA* := *array*([[0, 1], [−4, 0]])

$$AA := \begin{bmatrix} 0 & 1 \\ -4 & 0 \end{bmatrix} \quad \textbf{(9)}$$

> *BB* := *array*([0, 2·cos(t)])

$$BB := \begin{bmatrix} 0 & 2\cos(t) \end{bmatrix} \quad \textbf{(10)}$$

> *with*(*linalg*) :

> *MatExp* := *exponential*(*AA*, t)

$$MatExp := \begin{bmatrix} \cos(2t) & \frac{1}{2}\sin(2t) \\ -2\sin(2t) & \cos(2t) \end{bmatrix} \quad \textbf{(11)}$$

> $Xcero := array([0, 0])$

$$Xcero := \begin{bmatrix} 0 & 0 \end{bmatrix} \tag{12}$$

> $SolGral := evalm(MatExp \&* Xcero)$

$$SolGral := \begin{bmatrix} 0 & 0 \end{bmatrix} \tag{13}$$

> $MatExpTau := map(rcurry(eval, t='t - \text{tau}'), MatExp)$

$$MatExpTau := \begin{bmatrix} \cos(2\,t - 2\,\tau) & \frac{1}{2}\sin(2\,t - 2\,\tau) \\ -2\sin(2\,t - 2\,\tau) & \cos(2\,t - 2\,\tau) \end{bmatrix} \tag{14}$$

> $BBtau := map(rcurry(eval, t='\text{tau}'), BB)$

$$BBtau := \begin{bmatrix} 0 & 2\cos(\tau) \end{bmatrix} \tag{15}$$

> $AAtau := evalm(\ MatExpTau \&* BBtau)$

$$AAtau := \begin{bmatrix} \sin(2\,t - 2\,\tau)\cos(\tau) & 2\cos(2\,t - 2\,\tau)\cos(\tau) \end{bmatrix} \tag{16}$$

> $SolPart := map(int, AAtau, \text{tau} = 0\,..t) : y[1](t) = SolPart[1]; y[2](t) = SolPart[2]$

$$y_1(t) = -\frac{2}{3}\cos(2\,t) + \frac{2}{3}\cos(t)$$

$$y_2(t) = \frac{4}{3}\sin(2\,t) - \frac{2}{3}\sin(t) \tag{17}$$

> $CompUno := eval(subs(y[1](t) = SolPart[1], y[2](t) = SolPart[2], lhs(SistEcua[1]) - rhs(SistEcua[1]) = 0)\ )$

$$CompUno := 0 = 0 \tag{18}$$

> $CompDos := eval(subs(y[1](t) = SolPart[1], y[2](t) = SolPart[2], lhs(SistEcua[2]) - rhs(SistEcua[2]) = 0)\ )$

$$CompDos := 0 = 0 \tag{19}$$

> $plot([SolPart[1], SolPart[2]], t = 0\,..2\cdot\text{Pi})$



> *restart*

PREGUNTA 3 (30 puntos) Detremine una solución completa de la ecuación diferencial utilizando el método de separación de variables para una constante de separación nula (***sin usar pdsolve***)

> *Ecua* := *diff*(*y*(*x*, *t*), *t*\$2) + *diff*(*y*(*x*, *t*), *x*, *t*) = 2·*x*^3·*diff*(*y*(*x*, *t*), *t*)

$$Ecua := \frac{\partial^2}{\partial t^2} y(x,t) + \frac{\partial^2}{\partial x\,\partial t} y(x,t) = 2\,x^3 \left(\frac{\partial}{\partial t} y(x,t)\right) \quad \textbf{(20)}$$

> *EcuaSeparable* := *eval*(*subs*(*y*(*x*, *t*) = *F*(*x*)·*G*(*t*), *Ecua*))

$$EcuaSeparable := F(x)\left(\frac{d^2}{dt^2} G(t)\right) + \left(\frac{d}{dx} F(x)\right)\left(\frac{d}{dt} G(t)\right) = 2\,x^3\,F(x)\left(\frac{d}{dt} G(t)\right) \quad \textbf{(21)}$$

$$> EcuaSeparada := \frac{\left(lhs(EcuaSeparable) - \left(\frac{d}{dx} F(x)\right)\left(\frac{d}{dt} G(t)\right)\right)}{F(x)\cdot diff(\,G(t), t)} = simplify\left(\frac{\left(rhs(EcuaSeparable) - \left(\frac{d}{dx} F(x)\right)\left(\frac{d}{dt} G(t)\right)\right)}{F(x)\cdot diff(\,G(t), t)}\right)$$

$$EcuaSeparada := \frac{\frac{d^2}{dt^2} G(t)}{\frac{d}{dt} G(t)} = \frac{2\,F(x)\,x^3 - \left(\frac{d}{dx} F(x)\right)}{F(x)} \quad \textbf{(22)}$$

> *EcuaX* := *rhs*(*EcuaSeparada*) = 0

$$EcuaX := \frac{2\,F(x)\,x^3 - \left(\frac{d}{dx} F(x)\right)}{F(x)} = 0 \quad \textbf{(23)}$$

> $EcuaT := lhs(EcuaSeparada) = 0$

$$EcuaT := \frac{\frac{d^2}{dt^2} G(t)}{\frac{d}{dt} G(t)} = 0 \quad \textbf{(24)}$$

> $SolX := dsolve(EcuaX)$

$$SolX := F(x) = \_C1\, e^{\frac{1}{2} x^4} \quad \textbf{(25)}$$

> $SolT := dsolve(EcuaT)$

$$SolT := G(t) = \_C1\, t + \_C2 \quad \textbf{(26)}$$

> $SolGralCero := y(x, t) = rhs(SolT) \cdot subs(\_C1 = 1, rhs(SolX))$

$$SolGralCero := y(x, t) = (\_C1\, t + \_C2)\, e^{\frac{1}{2} x^4} \quad \textbf{(27)}$$

> $Comprobacion := simplify(eval(subs(y(x, t) = rhs(SolGralCero), lhs(Ecua) - rhs(Ecua) = 0)))$

$$Comprobacion := 0 = 0 \quad \textbf{(28)}$$

> $restart$

PREGUNTA 4 (30 puntos) Determinar la solución de la ecuación diferencial considerando una constante de separación positiva (***sin usar pdsolve***)

> $Ecua := diff(z(x, y), x\$2, y) = diff(z(x, y), x)$

$$Ecua := \frac{\partial^3}{\partial y\, \partial x^2} z(x, y) = \frac{\partial}{\partial x} z(x, y) \quad \textbf{(29)}$$

> $EcuaSeparable := eval(subs(z(x, y) = F(x) \cdot G(y), Ecua))$

$$EcuaSeparable := \left(\frac{d^2}{dx^2} F(x)\right) \left(\frac{d}{dy} G(y)\right) = \left(\frac{d}{dx} F(x)\right) G(y) \quad \textbf{(30)}$$

> $EcuaSeparada := \frac{lhs(EcuaSeparable)}{\left(\frac{d}{dx} F(x)\right) \cdot \left(\frac{d}{dy} G(y)\right)} = \frac{rhs(EcuaSeparable)}{\left(\frac{d}{dx} F(x)\right) \cdot \left(\frac{d}{dy} G(y)\right)}$

$$EcuaSeparada := \frac{\frac{d^2}{dx^2} F(x)}{\frac{d}{dx} F(x)} = \frac{G(y)}{\frac{d}{dy} G(y)} \quad \textbf{(31)}$$

> $EcuaX := lhs(EcuaSeparada) = \beta^2$

$$EcuaX := \frac{\frac{d^2}{dx^2} F(x)}{\frac{d}{dx} F(x)} = \beta^2 \quad \textbf{(32)}$$

> $EcuaY := rhs(EcuaSeparada) = \beta^2$

(33)

$$EcuaY := \frac{G(y)}{\frac{d}{dy} G(y)} = \beta^2 \tag{33}$$

> $SolX := dsolve(EcuaX)$

$$SolX := F(x) = \_C1 + \_C2\, e^{\beta^2 x} \tag{34}$$

> $SolY := dsolve(EcuaY)$

$$SolY := G(y) = \_C1\, e^{\frac{y}{\beta^2}} \tag{35}$$

> $SolGral := z(x, y) = subs(\_C1 = 1, rhs(SolY)) \cdot rhs(SolX)$

$$SolGral := z(x, y) = e^{\frac{y}{\beta^2}} \left(\_C1 + \_C2\, e^{\beta^2 x}\right) \tag{36}$$

> $Comp := eval(subs(z(x, y) = rhs(SolGral), lhs(Ecua) - rhs(Ecua) = 0))$

$$Comp := 0 = 0 \tag{37}$$

>

FIN DE LA SOLUCIÓN

>